

เทคนิคการใช้ Excel ในการวางแผนการผลิต (Excel Base Finite Scheduling System) - 2

ที่มา: www.production-scheduling.com

โดย: สารวย เปลื้องวัน (spluangwan@yahoo.com)

จากบทความตอนแรกผู้เขียนได้กล่าวถึงพื้นฐานที่จำเป็นในการสร้าง Finite Scheduling โดยใช้ Excel และได้กล่าวถึงการประยุกต์ใช้งาน สำหรับเครื่องจักรหนึ่งเครื่อง และหลายเครื่องทำงานขนานกัน ซึ่งครอบคลุม Section 1-12 ในโปรแกรมตัวอย่าง Scheduling_Excel.xls

ในตอนที่สองนี้จะกล่าวถึง การประยุกต์ใช้กับการทำงานแบบ Repetitive และการกำหนด Setup Matrix

Repetitive Production and Setups

จากตัวอย่างก่อนๆ เราจะกำหนดเวลาทำงานโดยกำหนดเป็นชั่วโมงทำงาน (Run Time) เช่น งาน A ใช้เวลา 7 ชั่วโมง ใน 7 ชั่วโมงคือเวลาทำงานทั้งหมดซึ่งจะรวมเวลาทำงานและเวลา Setup เครื่องจักรเข้าไว้ด้วยกัน กล่าวคือ

$$\text{Run Time} = \text{Setup Time} + \text{Processing Time}$$

แต่ถ้าลักษณะการทำงานเป็นการผลิตของซ้ำๆกัน (Repetitive Production) เช่น การฉีดพลาสติก, การผสมขนมปัง เวลาที่ใช้ในการ Setup เครื่องจักรจะคงที่ แต่ในส่วนของ Processing Time จะขึ้นอยู่กับจำนวนของที่ต้องผลิต โดยทั่วไปสามารถกำหนดได้สองแบบคือเป็น Cycle Time เช่นเวลาในการฉีดพลาสติกเป็น 2 นาที / Cycle (ในหนึ่ง Cycle อาจมีได้มากกว่า 1 ชิ้น) และอีกลักษณะหนึ่งคือกำหนดเป็น Production Rate เช่น 30 ชิ้น / ชม.

ตัวอย่างต่อไปจะเป็นการประยุกต์ใช้กับการวางแผนการผลิตกับเครื่องจักรที่ทำงานในลักษณะ Repetitive Production โดยจะกำหนด Setup Time และ Production Rate เอาไว้ เมื่อกำหนดงานและจำนวนลงในแผนการผลิต โปรแกรมจะทำการคำนวณ Run Time ให้โดยอัตโนมัติ

สมมุติว่ามีเครื่องจักรอยู่สองเครื่องคือ WC 1 และ WC 2 โดยมี Set Up Time คือ 1.25 และ 3 ชม. ตามลำดับ เครื่องจักรทั้งสองเครื่องสามารถผลิตงานได้ 4 แบบ โดยมี Production Rate ต่อชั่วโมงต่างกัน สร้าง Worksheet สำหรับ Setup Time และ Production Rate ได้ดังรูปที่ 1

	A	B	C	D	E	F	G
1							
2						Unit Per Hour	
3		W/C	Set Up Hours		Product	W/C 1	W/C 2
4		1	1.25		Prod A	80	140
5		2	3.00		Prod B	55	105
6					Prod C	75	135
7					Prod D	65	110

รูปที่ 1 : ตารางกำหนด Setup Time และ Production Rate

เวลา Run Time สามารถคำนวณได้จาก Setup Time + Processing Time โดยที่ Processing Time คือ Production Rate * Quantity สร้างตารางคำนวณ Run Time ดังรูปที่ 2 โดยกำหนด Column A สำหรับกำหนดเครื่องจักร, Column B กำหนดลำดับการทำงาน, Column C กำหนดงานที่จะทำ, Column D กำหนดจำนวนที่ต้องการ

ผลิต, Column E เป็นเวลา Setup เครื่องจักรให้ใ้สุด =VLOOKUP(A11,B\$4:C\$5,2) ลงใน E11, Function VLOOKUP เป็นการค้นหาค่า Setup Time ของเครื่องจักรที่ระบุใน A11 จาก B4:C5, Column F เป็น Production Rate ให้ใ้สุด =VLOOKUP(C11,E\$4:G\$7,A11+1) ซึ่งเป็นการค้นหาค่า Production Rate ของงานที่ระบุใน C11 จาก E4:G7, Column G เป็นเวลาที่ใ้ผลิตจริง =D11/F11, Column F เป็นเวลารวม ในหน่วยของชั่วโมง =E11+G11, Column G เป็นเวลารวมในหน่วยของวัน =H11/24, หลังจากกำหนดสูตรทั้งหมดแล้วก็ทำการ Copy ไปยัง Row อื่นๆ ดังรูป

เมื่อได้เวลา Run Time ใน Column G แล้วที่เหลือก็เข้า

	A	B	C	D	E	F	G	H	I
1									
2									
3		W/C	Set Up Hours			Unit Per Hour			
4		1	1.25			Product	W/C 1	W/C 2	
5		2	3.00			Prod A	80	140	
6						Prod B	55	105	
7						Prod C	75	135	
8						Prod D	65	110	
9	W/C	Seq.	Product	Qty	Set Up Hours	Unit Per Hour	Run Hours	Total Hours	Days
10									
11	1	1	Prod D	200	1.25	65	3.077	4.327	0.18
12	1	2	Prod B	350	1.25	55	6.364	7.614	0.317
13	1	3	Prod A	450	1.25	80	5.625	6.875	0.286
14	1	4	Prod B	300	1.25	55	5.455	6.705	0.279
15	2	1	Prod C	1500	3	135	11.11	14.11	0.588
16	2	2	Prod D	1100	3	110	10	13	0.542

รูปที่ 2 : การคำนวณ Run Time จาก ตาราง Setup และ Production Rate

สูตรการคำนวณ Start-Stop Time เหมือนในตัวอย่างก่อนๆ, ก่อนอื่นกำหนด Cell J4 ให้เป็น Cell พิเศษสำหรับกำหนดเวลาเริ่มงานของงานแรก ดังรูปที่ 3, Column J ใช้กำหนดเวลาเริ่มต้นของงานใ้สุด =IF(A11=A10,N10,J\$4) โดยที่ถ้าเป็นงานในเครื่องจักรเดียวกัน (A11=A10) จะใช้เวลาสิ้นสุดงานของงานก่อนหน้า (N10) เป็นเวลาเริ่มต้น ถ้าไม่ใ้จะใช้เวลาที่กำหนดใน J4

	J	K	L	M	N	
1						
2						
3		Job Start At				
4		2/28/03 8:30 AM				
5						
6						
7						
8						
9		Job Start	Calc 1	Calc 2	Calc 3	Job Stop
10						
11		2/28/03 8:30 AM	7	0.521	9	2/28/03 1:34 PM
12		2/28/03 1:34 PM	9	0.701	12	3/3/03 8:26 AM
13		3/3/03 8:26 AM	12	1.018	15	3/3/03 4:18 PM
14		3/3/03 4:18 PM	15	1.305	18	3/4/03 10:16 AM
15		2/28/03 8:30 AM	7	0.521	13	3/3/03 10:51 AM
16		3/3/03 10:51 AM	13	1.109	18	3/4/03 11:51 AM

รูปที่ 3 : การคำนวณ Job Start - Stop Time

ใน Column ที่เหลือใช้สูตรดังนี้

- Calc 1

=MATCH(J11,Cal_1_PB) + (MATCH(J11,Cal_1_PB) = MATCH(J11,Cal_1_PE))

- Calc 2

=INDEX(Cal_1_CWD,K11,1) - (INDEX(Cal_1_PE,K11,1) - MAX(INDEX(Cal_1_PB,K11,1),J11))

- Calc 3

=MATCH(L11+I11,Cal_1_CWD)+1

- Job Stop

=INDEX(Cal_1_PE,M11,1) - INDEX(Cal_1_CWD,M11,1)+L11+I11

หลักการทางานเหมือนกับที่ได้อธิบายไว้ในบทความตอนที่แล้ว โดย Cal_1_PB คือช่วงเวลา Period Begin ในปฏิทิน, Cal_1_PE คือช่วงเวลา Period End ในปฏิทิน, และ Cal_1_CWD คือช่วงเวลา Cumulative Work Day ในปฏิทิน

มีข้อควรระวังสำหรับการใช้งานคือสูตรหาค่าเวลาเริ่มต้นการทำงานใน Column J จะเปรียบเทียบเครื่องจักรกับเครื่องจักรใน Row ก่อนหน้า (A11=A10) ถ้าไม่ตรงกันก็จะตีความว่าเป็นงานแรก ดังนั้นถ้ามีการเพิ่มงานเข้าไปในแผนจะต้องมีการ Sort ตามเครื่องจักร (Column A) และ ลำดับการทำงาน (Column B) เสียก่อนเพื่อไม่ให้โปรแกรมทำงานผิดพลาด

จุดที่น่าสังเกตอีกอย่างคือผู้ใช้สามารถเพิ่มจำนวนเครื่องจักรและ Product เข้าไปได้ไม่จำกัด (ในทางทฤษฎี) จากรูปที่ 1 ตารางเครื่องจักรและ Setup สามารถเพิ่มต่อท้ายไปได้เรื่อยๆ โดยที่สูตรการหาเวลา Setup ใน Column E จะทำงานได้ไม่ผิดพลาดเพียงแต่ต้องขยายช่วงในสูตร VLOOKUP ให้ครอบคลุมถึงเครื่องจักรที่เพิ่มเข้าไป, ในขณะเดียวกัน ตรายใดที่ชื่อของเครื่องจักรยังเป็นหมายเลขที่เรียงลำดับกันและผู้ใช้ขยายช่วงในสูตร VLOOKUP ให้ครอบคลุมถึงเครื่องจักรและงานที่เพิ่มเข้าไป สูตรในการหาค่า Production Rate ใน Column F ก็ยังทำงานได้ไม่ผิดพลาดเช่นกัน

Setup Matrix

ตัวอย่างที่แล้วเวลา Setup Time ของเครื่องจักรเป็นเวลาดังที่ แต่ในบางกรณี Setup Time จะขึ้นอยู่กับงานก่อนหน้า เช่น เครื่องผสมสีเมื่อมีการผลิตสีใดสีหนึ่งเสร็จแล้วต้องมีการทำความสะอาดเครื่องก่อนที่จะทำการผสมสีใหม่ และเวลาที่ใช้ในการทำความสะอาดก็ขึ้นอยู่กับว่าสีที่เพิ่งผสมเสร็จไปและสีที่จะผสมใหม่นั้นเป็นสีอะไร ถ้าผสมสีขาวเสร็จและกำลังจะผสมสีดำต่อไปก็จะใช้เวลาในการทำความสะอาดนานกว่าถ้าจะผสมสีเทาต่อ เป็นต้น

การกำหนด Setup Time สำหรับการเปลี่ยนงานแต่ละชนิดสามารถทำได้โดยกำหนดให้อยู่ในรูปของ Matrix ดังรูปที่ 4 เครื่องจักรแต่ละเครื่องมี Setup Matrix แยกกันโดย Setup Time จะกำหนดอยู่ในช่องที่ตัดกันของ Row กับ Column เช่น เวลาที่เปลี่ยนจาก Product C ไปเป็น Product A ของเครื่องจักร 1 เป็น 2 ชม.

กำหนดชื่อให้ส่วนต่างๆของ Matrix โดยใช้ Menu Insert -> Name -> Define ...

ดังนี้

- Matrix_1 ช่วง A3:A7
- Matrix_1 ช่วง G3:K7
- Top_Row_1 ช่วง A3:E2
- Top_Row_2 ช่วง G3:K

กำหนดให้ G9 เป็นหมายเลขเครื่องจักร, G10 เป็นงานที่เพิ่งทำเสร็จ, และ G11 เป็นงานที่จะทำต่อไป กลไกการหาค่า Setup Time ประกอบด้วย 4 ขั้นตอนคือ

	A	B	C	D	E	F	G	H	I	J	K
1											
2		Work Center - 1 (Minute)					Work Center - 2 (Minute)				
3		To	Prod A	Prod B	Prod C	Prod D	To	Prod A	Prod B	Prod C	Prod D
4	Prod A	From	0.00	0.50	0.50	1.25	Prod A	0.00	1.25	1.25	2.75
5	Prod B		2.00	0.00	0.75	1.00	Prod B	4.25	0.00	1.75	2.50
6	Prod C		2.00	2.00	0.00	0.75	Prod C	4.25	4.25	0.00	1.75
7	Prod D		2.10	1.75	1.50	0.00	Prod D	5.00	4.50	3.50	0.00
8											
9					Work Center		1				
10					Previous Product		Prod D				
11					This Product		Prod B				
12											
13	Name Of Top Row						Top_Row_1				
14	Name Of Matrix						Matrix_1				
15	Position of This Product in Top Row						3				
16	Look Up Previous Product in Matrix						1.75				

รูปที่ 4 : การหาค่า Setup Time จาก Setup Matrix

1. ชื่อของช่วง Top Row: จะใช้การเชื่อม String โดยกำหนดสูตรต่อไปนี้ลงใน G13
="Top_Row_"&G9
2. ชื่อของ Setup Matrix: ใช้การเชื่อม String เช่นกัน โดยกำหนดสูตรต่อไปนี้ลงใน G14
="Matrix_"&G9
3. ตำแหน่งของงานปัจจุบันในช่วง Top Row: โดยการใช้ MATCH ร่วมกับการอ้างถึงช่วง Top Row แบบ Indirect โดยใช้สูตรต่อไปนี้ในช่อง G15
=MATCH(G11,INDIRECT(G13))
4. ค่าของ Setup Time ใน Matrix: เมื่อได้ตำแหน่งของงานปัจจุบันแล้วก็นำมาใช้ร่วมกับงานก่อนหน้าเพื่อหาค่าของ Setup Time โดยใช้สูตร
=VLOOKUP(G10,INDIRECT(G14),G15)

การทำงานทั้งหมดสามารถรวมกันให้เป็นสูตรเดียวได้ดังนี้
=VLOOKUP(G10, INDIRECT ("Matrix_"&G9), MATCH (G11, INDIRECT ("Top_Row_"&G9)))

การนำไปใช้งานให้ทดลองปรับเปลี่ยนตัวอย่างที่แล้วโดยเปลี่ยนสูตรใน Cell E11 ให้เป็น

=IF(A11=A10,VLOOKUP(C10,INDIRECT ("Matrix_"&A11),MATCH(C11,INDIRECT ("Top_Row_"&A11))),0)

จากนั้นก็ Copy ลงไปยัง Cell อื่นใน Column E เวลา Setup Time ก็จะเปลี่ยนไปใช้ตามที่กำหนดใน Setup Matrix ดังรูปที่ 5

มีข้อควรระวังสำหรับการใช้งานคือสูตรหาค่า Setup Time ใน Column E จะเปรียบเทียบเครื่องจักรกับเครื่องจักรใน Row ก่อนหน้า (A11=A10) ถ้าไม่ตรงกันก็จะตีความว่าเป็นงานแรกและจะกำหนด Setup Time = 0 ดังนั้นถ้ามีการเพิ่มงานเข้าไปในแผนจะต้องมีการ Sort ตามเครื่องจักร

(Column A) และ ลำดับการทำงาน (Column B) เสียก่อน เพื่อไม่ให้โปรแกรมทำงานผิดพลาด

	A	B	C	D	E
1					

g	W/C	Seq.	Product	Qty	Set Up Hours
10					
11	1	1	Prod D	200	0
12	1	2	Prod B	350	1.75
13	1	3	Prod A	450	2
14	1	4	Prod B	300	0.5
15	2	1	Prod C	1500	0
16	2	2	Prod D	1100	1.75

รูปที่ 5 : การใช้ Setup Matrix

สรุป

บทความตอนนี้ได้ยกตัวอย่างการประยุกต์ใช้กับการทำงานแบบ Repetitive ที่มีการกำหนดเวลาการผลิตแบบ Production Rate

เนื้อหาทั้งหมดครอบคลุม Section 13-15 ในโปรแกรมตัวอย่าง Scheduling_Excel.xls ซึ่งสามารถ Download ได้ที่ www.production-scheduling.com ถ้าต้องการโปรแกรมตัวอย่างของผู้เขียนก็สามารถ email มาขอได้ที่ spluangwan@yahoo.com โดยระบุ Subject ว่า Request Excel Sch - 2